



DECUS

PROGRAM LIBRARY

DECUS NO.	8-577a
TITLE	Paper Tape Duplicator (P.D.T.)
AUTHOR	D. Geoffrey Chase
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	February 23, 1974
SOURCE LANGUAGE	PAL III

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

21750

1914



PAPER TAPE DUPLICATOR (P.D.T.)

Revision of 2/74

Purpose: To perform many of the functions of the DEC Master Tape Dupli-
tor without using the interrupt facility.

Hardware: PDP-8 series processor, high-speed paper tape reader & punch.
The program can be modified for low-speed punch and even for
the low-speed reader.

Core: 10,11,20-161,176-431 plus storage buffer 600-7577, in any core
field. Since interrupt is not used, this should run under TSS-8.

Input Format: Virtually any kind of paper tape can be copied. The tape
should end in a square (not forked) tail if the high-speed reader
is in use. Two or more successive rubouts are considered to
mark the end of the tape.

Output Format:

- (a) leader code, 63 frames of code 200 or of blanks (see below)
- (b) the original program with its trailer, copied "as is"
- (c) three Rubouts (original master tape) or two (copies)
- (d) Ctrl/Z (end of file), Ctrl/FORM (end of page)
- (e) a 24-bit straight checksum of all frames except the leader
- (f) a 24-bit circular checksum with end-around carry, to catch
a punch that both drops and adds bits in the same channel
- (g) a short section of trailer.

Ignored: (a) Leader, blank or code 200, until something else has been read.
(b) Rubouts immediately following 2 successive rubouts.

Switch Register Options:

- (a) If bit 0 is set, a new master tape is created.
- (b) If bit 0 is clear, bit 1 set, a master or its copy is copied;
the input tape is verified against its checksum.
- (c) If bits 0 & 1 are both clear, a master or its copy is simply
read and verified against its checksum (no duplication).
- (d) Bit 4, if set, specifies code 200 leader/trailer; if clear,
code 000 (blanks). RIM & BIN tapes must have code 200; ASCII
(source) tapes may have either type of leader.

Teletype Messages to User: none.

Halts:

- (a) At address 176: normal, no errors.
- (b) At address 44: reader timed out. This has never happened here;
it was included as a precaution against jams.
- (c) At address 426: bad checksum on input tape. See listing.

Notes:

- (a) The checksum material is punched with parity set (channel 8 always punched). If the tape is a RIM tape, the RIM loader will ignore this, as it does any code 200 or higher.
- (b) The combination of Ctrl/Z and FORM should stop input to almost any editor or OS-8 program.
- (c) Copies of binary tapes will presumably halt on the copy of the original code 200 trailer.
- (d) Source tapes for assemblers will have a \$ or END, which will cause the assembler to ignore all that follows.

In brief, the checksum material should prove relatively innocuous. To date, no tape that verifies has proved bad.

Changing Devices:

- (a) Switching to the low-speed punch is relatively easy. Change the 602x codes in 65,75,200,322,324 to the corresponding 604x codes.
- (b) The reader is another matter, because of the 25-millisecond timing loop. Your options are to change the timing loop--probably an off-page jump to, say, loc. 432 will be needed--or to take it out and count on typing 2 rubouts by hand when end of input tape is reached.

Use:

- (a) Load with binary loader into any core field.
- (b) Load starting address = 0200 in that field.
- (c) Choose your switch register options--see previous page.
- (d) Check your reader and punch. Both should be on, the input tape in the reader ready to go.
- (e) Press Clear and Continue ['Start' on earlier PDP-8's].

Restart:

- (a) From normal halt and checksum error halt, just press Continue.
- (b) If a time-out halt is due to end of tape, press Continue to finish job. The HLT in location 44 was put in because, in fact, the writer has never seen a tape time out--the tail end gets read by the H.S.R. as a lot of rubouts.

This version is compatible with its predecessors. The principal changes: (a) larger text buffer (b) the bit 4 option (c) removal of instructions peculiar to the PDP-8/E and later models (d) reformatting of the listing.

For those whose high-speed readers consistently halt (loc. 44) on time-out rather than on multiple rubouts, perhaps the following is better than the suggestion made on the first page of the listing:
Into loc. 44, deposit 7040; into loc. 45, deposit 1135 [simulates rubout.

/PAPER TAPE DUPLICATOR (P.D.T.):
/FOR PDP-8 SERIES COMPUTERS, H. S. R./H. S. P.

/REV. 2/23/74 PAS PAGE 0 ROUTINES &C.:

*20

/0-7, 12-17 LEFT FREE

0020	0000	EXIT,	0	/POINTER BACK TO MAIN PROGRAM
0021	0000	CTR2,	0	
0022	0000	RBTF LG,	0	/-1 AFTER RUBOUT
0023	0577	BOTM,	577	/BUFFER: 600 TO 7577
0024	0000	SAVE,	0	
		DMY=	ZNOR	

0025	7200	READP,	CLA	/READ & PUNCH ROUTINE:
0026	0000	SWI,	0	/WILL HOLD 0 OR JMP PUN
0027	1010		TAD 10	
0030	1133		TAD K201	/[UPPER STORAGE LIMIT=7577]
0031	7640		SZA CLA	/STORAGE FULL
0032	5036		JMP TIME	/ROOM STILL IN STORAGE AREA

0033	1035		TAD KJMPI	
0034	3026		DCA SWI	
0035	5064	KJMPI,	JMP PUN	/STORAGE FULL. SKIP READER

0036	3101	TIME,	DCA DMY	
0037	6011		RSF	
0040	7410		SKP	
0041	5046		JMP READIT	
0042	2101		ISZ DMY	
0043	5037		JMP .-4	/ABOUT 25 MILLISECS. (8/E)
0044	7402		HLT	/[OR NOP]
0045	5420		JMP I EXIT	

/AUTHOR'S READER TRIPS OUT ON RUBOUTS AFTER END OF TAPE,
/NOT ON TIMING. IF YOURS ACTS OTHERWISE, NOP THE 'HLT'.

0046	6016	READIT,	RRB RFC	
0047	3024		DCA SAVE	
0050	4536		JMS I PCHECK	/STORE, ADD TO CHECKSUMS
0051	1024		TAD SAVE	
0052	1176		TAD HALT	/RUBOUT?
0053	7740		SMA CLA SZA	
0054	5057		JMP .+3	/YES
0055	3022		DCA RBTF LG	/NO. CLEAR FLAG.
0056	5064		JMP PUN	

/.....

0057	2022		ISZ RBTF LG	/WAS LAST CHAR. A RUBOUT?
0060	7410		SKP	
0061	5420		JMP I EXIT	/YES, 2 IN A ROW
0062	7040		CMA	/NO, BUT SET FLAG = -1
0063	3022		DCA RBTF LG	

/.....

0064	0000	PUN,	0	/0 OR JMP READP
0065	6021		PSF	
0066	5025	KJMP2,	JMP READP	/LOOP UNTIL END OF TAPE
0067	1010		TAD 10	/PUNCH CAUGHT UP W. READER?
0070	7041		CIA	
0071	1011		TAD 11	
0072	7650		SNA CLA	
0073	5077		JMP CGHTUP	/YES
0074	1411		TAD I 11	/NO, PUNCH STORED CHAR.
0075	6026		PLS	
0076	5025		JMP READP	/LOOP UNTIL END OF TAPE
0077	4137	CGHTUP,	JMS RESET	
0100	5025		JMP READP	/READER ENABLED
0101	0000	ZNOR,	0	/SUBR. TO READ LEADER
0102	4107		JMS READ	
0103	0132		AND K177	/IGNORE BLANKS, 200'S
0104	7650		SNA CLA	
0105	5102		JMP --3	
0106	5501		JMP I ZNOR	/EXIT WITH FRAME IN SAVE
0107	0000	READ,	0	/READS LEADER; CHECKSUMS
0110	6011		RSF	
0111	5110		JMP --1	
0112	7200	KCLA,	CLA	
0113	6016		RRB RFC	
0114	3024		DCA SAVE	
0115	1024		TAD SAVE	
0116	5507		JMP I READ	
0117	0000	ADCHK,	0	/SUBR. TO READ TAPE CKSUMS.
0120	4107		JMS READ	/GET FRAME
0121	1176		TAD HALT	
0122	7740		SMA SZA CLA	
0123	5120		JMP --3	/IGNORE RUBOUTS, CTRL/Z,
0124	4107		JMS READ	/...AND FORM.
0125	4535		JMS I GET2	/GET 2-FRAME CKSUM WORDS
0126	0000	CK1A,	0	/ARGUMENTS (4)
0127	0000	CK1B,	0	
0130	0000	CK2A,	0	
0131	0000	CK2B,	0	/EXIT DIRECT (SEE "GETTWO")
0132	0177	K177,	177	
0133	0201	K201,	201	
0134	0077	K77,	77	
0135	0400	GET2,	GETTWO	
0136	0333	PCHECK,	CHECK	

0137	0000	RESET,	0	/RESET BUFFER PTRS., READER SWITCH
0140	7300		CLA CLL	
0141	1023		TAD BOTM	/577+1=600
0142	3010		DCA 10	
0143	1010		TAD 10	
0144	3011		DCA 11	
0145	3026		DCA SW1	
0146	5537		JMP I RESET	
0147	0000	LDR,	0	/SUBR. TO PUNCH LEADER
0150	7041		CIA	
0151	3021		DCA CTR2	
0152	7604		LAS	/READ BIT 4 OF S.R.
0153	0112		AND KCLA	/RESULT: BLANK, OR 200
0154	4560		JMS I .+4	/TO "PUNCH"
0155	2021		ISZ CTR2	
0156	5152		JMP .-4	
0157	5547		JMP I LDR	
0160	0321		PUNCH	
0161	0015	K15,	15	
*176				
0176	7402	HALT,	HLT	/NORMAL HALT, NO ERRORS.
0177	7610		CLA SKP	/RESTART BY "CONTINUE"
/**** START AT 0200: ****				
0200	6026		PLS	
0201	4137	RESTR,	JMS RESET	/BUFFER PTRS., READER
0202	6014		RFC	
0203	1066		TAD KJMP2	
0204	3064		DCA PUN	/INHIBIT PUNCH
0205	3126		DCA CK1A	
0206	3127		DCA CK1B	
0207	3130		DCA CK2A	
0210	3131		DCA CK2B	/ CLEAR 'EM
0211	7604		LAS	
0212	7710		SPA CLA	/BIT 0 SET?
0213	5237		JMP MASTER	/YES: "MASTER"
0214	7604		LAS	/NO, WAS BIT 1 SET?
0215	7104		CLL RAL	
0216	7710		SPA CLA	
0217	5232		JMP COPY	/YES: "COPY"
0220	1227	VERIFY,	TAD KEXITV	/NO, VERIFY A TAPE:
0221	3020		DCA EXIT	
0222	4101		JMS ZNOR	/IGNORE LEADER (BLANK OR 200)
0223	4333		JMS CHECK	/1ST SIGNIF. CHAR.
0224	5025		JMP READP	/PUNCH IS STILL INHIBITED
0225	4117	EXITV,	JMS ADCHK	/COMPUTE CHECKSUMS & COMPARE
0226	5176		JMP HALT	/IF "ADCHK" FINDS NO ERROR

0227	0225	KEXITV,	EXITV	
0230	0234	KEXITC,	EXITC	
0231	0247	KEXITM,	EXITM	
0232	1230	COPY,	TAD KEXITC	/COPY A MASTER TAPE:
0233	5240		JMP MASTER+1	
0234	4117	EXITC,	JMS ADCHK	
0235	4271		JMS PUNREST	
0236	5252		JMP PUNCHK+2	
0237	1231	MASTER,	TAD KEXITM	/CREATE A MASTER TAPE:
0240	3020		DCA EXIT	
0241	3064		DCA PUN	/ENABLE PUNCH
0242	1134		TAD K77	
0243	4147		JMS LDR	
0244	4101		JMS ZNOR	/EXIT W. NON-0 CHAR. IN SAVE
0245	4333		JMS CHECK	
0246	5025		JMP READP	
0247	4271	EXITM,	JMS PUNREST	
0250	1331	PUNCHK,	TAD RBT	/PUNCH RBTS., HALTS, CKSUMS
0251	4321		JMS PUNCH	/(MASTER GETS 3 RBTS., COPIES 2)
0252	1327		TAD CTRLZ	/1Z, END-OF-FILE FOR OS-8
0253	4321		JMS PUNCH	
0254	1330		TAD FORM	/END-OF-FILE FOR SYMBOLIC ED.
			/	(ED. IGNORES 1Z)
0255	4321		JMS PUNCH	
0256	1126		TAD CK1A	/PUNCH 8 CKSUM. FRAMES:
0257	4303		JMS PUN2	/1 WD. = 2 FRAMES
0260	1127		TAD CK1B	
0261	4303		JMS PUN2	
0262	1130		TAD CK2A	
0263	4303		JMS PUN2	
0264	1131		TAD CK2B	
0265	4303		JMS PUN2	
0266	1161		TAD K15	
0267	4147		JMS LDR	
0270	5176		JMP HALT	
0271	0000	PUNRES,	0	/PUNCH REST OF TXT. BUFFER
0272	7200		CLA	
0273	1010		TAD 10	
0274	7041		CIA	
0275	1011		TAD 11	
0276	7650		SNA CLA	
0277	5671		JMP I PUNREST	/BUFFER EMPTY
0300	1411		TAD I 11	
0301	4321		JMS PUNCH	/PUNCH CHAR.
0302	5272		JMP PUNRES+1	

0303	0000	PUN2,	0	/PUNCH 1 WD. AS 2 FRAMES
0304	3332		DCA STOR	
0305	1332		TAD STOR	
0306	7112		CLL RTR	
0307	7012		RTR	
0310	7012		RTR	/FIRST HALF
0311	0134		AND K77	
0312	1112		TAD KCLA	/SET TAPE CHAN. 8
0313	4321		JMS PUNCH	
0314	1332		TAD STOR	
0315	0134		AND K77	
0316	1112		TAD KCLA	/SAME. (THIS TO FOOL RIM)
0317	4321		JMS PUNCH	
0320	5703		JMP I PUN2	
0321	0000	PUNCH,	0	
0322	6021		PSF	
0323	5322		JMP --1	
0324	6026		PLS	
0325	7200		CLA	
0326	5721		JMP I PUNCH	
0327	0232	CTRLZ,	232	
0330	0214	FORM,	214	
0331	0377	RBT,	377	
0332	0000	STOR,	0	
0333	0000	CHECK,	0	/STORE CHAR., UPDATE CHECKSUMS
0334	7300		CLA CLL	
0335	1020		TAD EXIT	
0336	7041		CIA	
0337	1227		TAD KEXITV	/ "VERIFY" PROGRAM?
0340	7650		SNA CLA	
0341	5344		JMP .+3	/YES, STORE NOTHING!
0342	1024		TAD SAVE	/NO, STORE CHAR. IN BUFFER
0343	3410		DCA I 10	
/ "STRAIGHT" CKSUM:				
0344	7100		CLL	
0345	1024		TAD SAVE	
0346	1127		TAD CK1B	
0347	3127		DCA CK1B	
0350	7004		RAL	/GET CARRY
0351	1126		TAD CK1A	
0352	3126		DCA CK1A	
/ "CIRCULAR" CKSUM:				
0353	1131		TAD CK2B	
0354	7104		CLL RAL	
0355	3131		DCA CK2B	
0356	1130		TAD CK2A	
0357	7004		RAL	
0360	3130		DCA CK2A	

0361	7430		SZL	
0362	2131		ISZ CK2B	/WHICH CAN'T OVERFLOW
0363	7100		CLL	/NOW ADD NEW FRAME
0364	1024		TAD SAVE	
0365	1131		TAD CK2B	
0366	3131		DCA CK2B	
0367	7004		RAL	/GET CARRY
0370	1130		TAD CK2A	
0371	3130		DCA CK2A	
0372	7420		SNL	
0373	5733		JMP I CHECK	/NO OVERFLOW
		/.....		
0374	2131		ISZ CK2B	
0375	5733		JMP I CHECK	
0376	2130		ISZ CK2A	/IF THERE WAS A CARRY
0377	5733		JMP I CHECK	
		/.....		
0400	0000	GETTWO, 0		/READ 2 24-BIT CHECKSUMS (8 FRAMES); /COMPARE WITH COMPUTED "CK" VALUES.
0401	7200		CLA	
0402	1230		TAD MIN4	
0403	3021		DCA CTR2	
0404	4107	LOOPG,	JMS READ	/GET FRAME
0405	0134		AND K77	/DROP THE EDGE PUNCH (200)
0406	7106		CLL RTL	
0407	7006		RTL	
0410	7006		RTL	
0411	3231		DCA TEMP	/FRAME IN BITS 0-5 OF TEMP
0412	4107		JMS READ	
0413	0134		AND K77	
0414	1231		TAD TEMP	/PACK INTO 1 WD.
0415	7041		CIA	
0416	1600		TAD I GETTWO	/ARGUMENTS: 4 SUCCESSIVE WDS. / [CK1A,CK1B,CK2A,CK2B]
0417	7640		SZA CLA	
0420	5225		JMP ERROR	/DISCREPANCY BETWN. CKSUMS.
0421	2200		ISZ GETTWO	/POINT TO NEXT ARG.
0422	2021		ISZ CTR2	/DONE 8 FRAMES YET?
0423	5204		JMP LOOPG	/NO
0424	5517		JMP I ADCHK	/YES. NO ERRORS.
0425	1200	ERROR,	TAD GETTWO	/SHOWS ADRS. OF CKSUM WHICH
0426	7402		HLT	/ DISAGREES.
0427	5177		JMP 177	/RESTART BY "CONTINUE"
0430	7774	MIN4,	-4	
0431	0000	TEMP,	0	

ADCHK	0117
BOTM	0023
CGHTUP	0077
CHECK	0333
CK1A	0126
CK1B	0127
CK2A	0130
CK2B	0131
COPY	0232
CTRLZ	0327
CTR2	0021
DMY	0000
ERROR	0425
EXIT	0020
EXITC	0234
EXITM	0247
EXITV	0225
FORM	0330
GETTWO	0400
GET2	0135
HALT	0176
KCLA	0112
KEXITC	0230
KEXITM	0231
KEXITV	0227
KJMP1	0035
KJMP2	0066
K15	0161
K177	0132
K201	0133
K77	0134
LDR	0147
LOOPG	0404
MASTER	0237
MIN4	0430
PCHECK	0136
PUN	0064
PUNCH	0321
PUNCHK	0250
PUNRES	0271
PUN2	0303
RBT	0331
RBTF LG	0022
READ	0107
READIT	0046
READP	0025
RESET	0137
RESTR T	0201
SAVE	0024
STOR	0332
SWI	0026
TEMP	0431
TIME	0036
VERIFY	0220
ZNOR	0101

